PODPROGRAMY (procedury, funkcje) **:**

- skrócenie zapisu *(wielokrotne powtórzenia sekwencji instrukcji)*,
- lepsze wykorzystanie pamięci operacyjnej *(zmienne lokalne)*,
- ułatwienie i przyśpieszenie projektowania programu *(dekompozycja)*,
- typowe operacje → biblioteki podprogramów,
- czytelność programu i łatwość modyfikacji,

Procedury i funkcje standardowe Pascala:

Moduły: *System, Crt, Dos, WinDos, Graph, Overlay, Strings*.
dołączenie: **uses** Crt, Graph;

INSTRUKCJA WYWOŁANIA PROCEDURY:

Nazwa_procedury **(** *Param_1***,** *Param_2***,** *. . .* **,** *Param_n* **);**

procedury bez parametrów *(np. ClrScr, Break, Continue)*,
procedury o zmiennej liczbie i zmiennym typie parametrów *(np. Write, Readln)*

Przekazywanie parametrów:

- przez wartość *(przy wywołaniu: zmienna, stała lub wyrażenie)*,
- przez zmienną → **var** *(przy wywołaniu: zmienna)*

np. def.:

procedure Insert (Source: **string**; **var** S: **string**; Index : **Integer**);

INSTRUKCJA WYWOŁANIA FUNKCJI:

- w wyrażeniu po prawej stronie instrukcji przypisania,
- w wyrażeniu będącym parametrem procedury lub funkcji

$$\mathbf{c} = \sqrt{\mathbf{a}^2 + \mathbf{b}^2} \quad \rightarrow \quad \mathbf{c} := \text{Sqrt}(\text{Sqr}(\mathbf{a}) + \text{Sqr}(\mathbf{b}));$$

np. def.:

function Copy (S: **string**; Index : Integer; Count: Integer) : **string**

```
program Kwadraty_liczb;
uses Crt;                 {dołączenie biblioteki}
var
   X: Integer;  R : Real ;
begin
   ClrScr;                {wywołanie procedury bez parametrów}
   X := 0;
   repeat
      Inc(X, 2) ;         {wyw. procedury, pierwszy parametr "przez zmienną"}
      R := Sqrt(X) ;      {wywołanie funkcji}
      Writeln(R) ;        {wywołanie procedury}
   until X = 20 ;
   Readln ;               {wywołanie procedury bez parametrów}
end.
```

DEFINIOWANIE PROCEDURY:

```
procedure Nazwa_procedury( lista_parametrów );
var  {itd., jak w sekcji deklaracji programu.
        mogą tu być również definicje: stałych, typów
        oraz wewnętrznych procedur i funkcji !}
begin
     {ciąg instrukcji procedury}
end ;
```

DEFINIOWANIE FUNKCJI:

```
function Nazwa_funkcji( lista_parametrów ): typ_wyniku;
var  {itd., jak w sekcji deklaracji programu.
        mogą tu być również definicje: stałych, typów
        oraz wewnętrznych procedur i funkcji !}
begin
     {ciąg instrukcji procedury}
     Nazwa_funkcji := obliczona_wartość ;
end ;
```

```
program Obliczenia_z_procedurami;

uses Crt;
var A,B,C,S  : Integer;
             Sr : Real;


procedure WCZYTAJ_DANE( var P1, P2, P3 : Integer );
   begin
     Write( 'Podaj 3 liczby całkowite: ' );
     Readln( P1, P2, P3 ) ;
   end ; {WCZYTAJ_DANE}


procedure POLICZ( P1, P2, P3:Integer;
                          var Suma: Integer;   var Srednia: Real );
   begin
     Suma := P1 + P2 + P3 ;
     Srednia := Suma / 3 ;
   end ; {POLICZ}


procedure WYSWIETL_WYNIKI( Suma : Integer; Srednia : Real );
   begin
     ClrScr;
     Writeln ( 'Suma = ', Suma);
     Writeln ( 'Srednia arytmetyczna = ', Srednia);
   end; {WYSWIETL_WYNIKI}


{program główny}

begin
   Wczytaj_dane( A, B, C );
   Policz( A, B, C, S, Sr ) ;
   Wyswietl_wyniki( S, Sr );
   Readln;
end.
```

```
program Obliczenia_z_funkcjami;

uses Crt;
var A,B,C,S  : Integer;
             Sr : Real;

procedure WCZYTAJ_DANE( var P1, P2, P3 : Integer );
   begin
     Write( 'Podaj 3 liczby całkowite: ' );
     Readln( P1, P2, P3 ) ;
   end ; {----------------------------------------------------------WCZYTAJ_DANE}


procedure WYSWIETL_WYNIKI( Suma : Integer; Srednia : Real );
   begin
     ClrScr;
     Writeln ( 'Suma = ', Suma);
     Writeln ( 'Srednia arytmetyczna = ', Srednia);
   end; {----------------------------------------------------------WYSWIETL_WYNIKI}


function  SUMA ( P1,P2,P3 : Integer) : Integer;
   begin
     Suma := P1 + P2 + P3;
   end ; {-------------------------------------------------------------------------SUMA}


function SREDNIA ( P1, P2, P3 : Integer ) : Real;
     var D : Integer;
   begin
     D := SUMA( P1, P2, P3 );
     Srednia := D / 3 ;
   end ; {-----------------------------------------------------------------------SREDNIA}

{ program główny }

begin
   Wczytaj_dane( A, B, C );
   S  := Suma( A, B, C ) ;
   Sr := Srednia( A, B, C ) ;
   Wyswietl_wyniki( S, Sr );
   Readln;
end.
```